[Touch Cruise 2 General] igor7:
Nice finding *THUMBS UP* as you like to say!! You are right! It working fine after removing file attributes. I also find this... Read my post:

But some where I was wrong to...

Anyway, I think you need to add option to remove file attributes to your original version of script. In this way it will always work properly. Here is corrected code (I added this line: attrib -r -s Manila2D.exe into your script. It remove read only and system attributes from Manila2D.exe):

@ECHO off
ECHO ============================================================
ECHO N2A's Automatic Weather Patched Manila2D.exe Package Creator
ECHO Created by N2A @ XDA-Developers.com
ECHO For all questions, please see the thread at XDA-Developers
ECHO and the site at http://weather.not2advanced.com/files/
ECHO ============================================================
ECHO.

IF EXIST Manila2D.exe GOTO htchomeok
ECHO Manila2D.exe was not found! Please place your Manila2D.exe file in the same folder as this one.
GOTO end

:htchomeok
IF EXIST tools\unpatched.dll del tools\unpatched.dll
attrib -r -s Manila2D.exe
copy Manila2D.exe tools\unpatched.dll > NUL
IF EXIST *.cab del *.cab
cd tools

REM ==============================
REM Make sure file isn't read only
REM ==============================
attrib -r -h -s +a unpatched.dll

REM ========================
REM Test for UPX compression
REM ========================
grep -U "packed with the UPX" unpatched.dll > NUL
IF ERRORLEVEL 1 GOTO postdecompress

REM ======================
REM File is UPX Compressed
REM ======================
ECHO UPX compressed file detected. Decompressing...
upx -q -d unpatched.dll > NUL

:postdecompress
REM ===================================
REM Patching file to use the N2A Server
REM ===================================
ECHO Performing patch...
binmay -s "68 00 74 00 74 00 70 00 3A 00 2F 00 2F 00 68 00 74 00 63 00 2E 00 61 00 63 00 63 00 75 00 77 00 65 00 61 00 74 00 68 00 65 00 72 00 2E 00 63 00 6F 00 6D 00 2F 00 77 00 69 00 64 00 67 00 65 00 74 00 2F 00 68 00 74 00 63 00 2F 00 66 00 6F 00 72 00 65 00 63 00 61 00 73 00 74 00 2D 00 64 00 61 00 74 00 61 00 5F 00 76 00 33 00 2E 00 61 00 73 00 70 00 3F 00 61 00 63 00 3D 00 54 00 52 00 32 00 63 00 72 00 61 00 39 00 55 00 26 00 6C 00 6F 00 63 00 43 00 6F 00 64 00 65 00 3D" -r "68 00 74 00 74 00 70 00 3A 00 2F 00 2F 00 77 00 65 00 61 00 74 00 68 00 65 00 72 00 2E 00 6E 00 6F 00 74 00 32 00 61 00 64 00 76 00 61 00 6E 00 63 00 65 00 64 00 2E 00 63 00 6F 00 6D 00 2F 00 68 00 74 00 63 00 77 00 65 00 61 00 74 00 68 00 65 00 72 00 2F 00 66 00 6F 00 72 00 65 00 63 00 61 00 73 00 74 00 64 00 61 00 74 00 61 00 2E 00 70 00 68 00 70 00 3F 00 61 00 63 00 3D 00 58 00 44 00 41 00 44 00 65 00 76 00 73 00 26 00 6C 00 6F 00 63 00 43 00 6F 00 64 00 65 00 3D" -i unpatched.dll -o patched.dll > NUL

REM ================================
REM Check that it's actually changed
REM ================================
bincomp unpatched.dll patched.dll /Q
IF ERRORLEVEL 3 GOTO patchedok
GOTO failedpatch

:patchedok
REM =============================
REM Re-signing file with SignCode
REM =============================
ECHO Re-signing file...
signcode -spc TestCert_Privileged.cer -v TestCert_Privileged.pvk patched.dll > NUL

REM =====================
REM Packing file with UPX
REM =====================
ECHO Packing file...
upx -q --best --compress-icons=0 patched.dll > NUL

REM ================================
REM Renaming and packing the package
REM ================================
ECHO Preparing cabinet install...
ren patched.dll Manila2D.exe
cabwiz inputinf.inf
move inputinf.cab "..\N2A Patched Manila2D EXE.cab"
del inputinf.dat
del unpatched.dll
del Manila2D.exe
cd ..
ECHO.
ECHO =======================================================================
ECHO Process is complete, and was successful!
ECHO Install the cabfile which has been placed in this folder to your phone.
ECHO =======================================================================
goto end

:failedpatch
del unpatched.dll
del patched.dll
cd ..
ECHO =======================================================================
ECHO Process failed! Unable to patch the Manila2D.exe file correctly.
ECHO Either this file has already been patched, or something has gone wrong.
ECHO If you are sure the file is not pre-patched, please report this error
ECHO on the XDA-Developers forums, using the link from
ECHO http://weather.not2advanced.com/files/
ECHO =======================================================================
:end
ECHO.
ECHO Press a key to exit.
PAUSE > NUL

Now this script working even under Windows 7: